
LOAN/SAVE 
Display 

  

(HP-41CX, Hewlett Packard 1983 and DM41X, SwissMicros 2020) 

 

Overview1 

Program L/S (with subprograms LOAN and SAVE) calculates monthly payments or savings based on an 
amount, an interest value and a period. The loan and savings program was written to review different 
ways of managing the mortgage and savings. Most banks can provide a nice listing but what if the interest 
changes? Below program may be of help. A feeling of nostalgia will be experienced when calculating a 
mortgage over a period of 30 years. The calculation loop has to be performed 360 times. The cursor has 
to move 30 times over the display which is shown by a year-counter. 

 

Calculations 

The loan on the basis of annuity is calculated by below algorithm in which AMT and MP represent the 
amount respectively monthly payment and r is the monthly interest and m the number of months: 

𝑀𝑃 = 𝐴𝑀𝑇 ∙  
𝑟𝑚

𝑟𝑚−1 + 𝑟𝑚−2 +. . + 𝑟 +  1
  

𝑀𝑃 =  𝐴𝑀𝑇 ∙  
𝑟𝑚

𝑟 ∙ (𝑟 ∙ (𝑟 ∙ (. . )  +  1)  + 1) +  1
 

 

Savings calculation is based on below algorithm in which SUM and MP represent the lump sum 
respectively monthly payment: 

𝑆𝑈𝑀 = 𝑟 ∙ (𝑟 ∙ (𝑟 ∙ (. . )  +  𝑀𝑃) + 𝑀𝑃) +  𝑀𝑃 

 

in which r expresses the monthly interest derived from the annual interest I (in %) as follows: 

𝑟 = 1 + 
𝐈%

 12 ∙ 100
 

 

  

  

 
1 This program is copyright and is supplied without representation or warranty of any kind. The author assumes no responsibility and shall have no 
liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof 

https://www.swissmicros.com/

Example (1): LOAN 

KEYSTROKES DISPLAY COMMENTS 

  Run L/S (main entry) 

[XEQ] [ALPHA]L/S[ALPHA]  Select L for loan 

L[R/S]  Enter the loan AMOUNT 

50000[R/S]  Enter the annual interest 

7,5[R/S]  Enter the duration in number of months 

240[R/S]  Count down in years.. 

   

   

   

  Monthly payment 

[XEQ] [ALPHA]LOAN[ALPHA]  Run LOAN directly 

12000[R/S]  Enter the annual interest 

2[R/S]  Enter the duration in number of months 

36[R/S]  Count down in years.. 

   

   

  Monthly payment 

[R/S]  Run again from main entry 

 

 

Example (2): SAVE 

KEYSTROKES DISPLAY COMMENTS 

  Run L/S (main entry) 

[XEQ] [ALPHA]L/S[ALPHA]  Select S for savings 

S[R/S]  Enter the monthly payment 

150[R/S]  Enter the annual interest 

3[R/S]  Enter the duration in number of months 

60[R/S]  Count down in years.. 

   

   

  Lump sum savings 

[CLX]  Clear display 

[XEQ] [ALPHA]SAVE[ALPHA]  Run SAVE directly 

999[R/S]  Enter the annual interest 

1,5[R/S]  Enter the duration in number of months 

24[R/S]  Count down in years.. 

   

  Lump sum savings 

[R/S]  Run again from main entry 

Program Listing 

The LOAN program’s outcome will be the Monthly Payment for the loan (mortgage) on the basis of 
annuity and requires the following parameters: 

▪ the Amount to borrow in your own currency 
▪ the effective Annual Interest 
▪ the Number of Months to pay it back 

 

The SAVE program calculates the final Lump Sum you receive after the period of saving and requires the 
following parameters: 

▪ the Monthly Payment in your own currency 
▪ the effective Annual Interest 
▪ the Number of Months to save your money 

 

The listing below shows an extended command in line 08: ATOX. This is to check whether the input 
character was an 'L' (value 76). For the C and CV this can be replaced by changing it an ASTO X followed 
by 'L'. The same applies to line 11 where the compare character is an 'S' (value 83). 

The number of bytes can easily be reduced by removing subroutine LBL 06 and its calls XEQ 06 or by 
deleting lines 01-13. 

The listing of program L/S (with subprograms LOAN and SAVE) is given below: 

 

 01▪LBL "L/S"  

 02▪LBL 00  

 03 "L/S?" 
 04 AON 

 05 PROMPT 

 06 AOFF  

 07 76  

 08 ATOX  

 09 X=Y?  
 10 GTO "LOAN" 

 11 83 

 12 X#Y? 

 13 GTO 00 

 14▪LBL "SAVE"  

 15 "M.P=?" 
 16 XEQ 05  

 17 *  

 18 "NR.M=?" 

 19 PROMPT  

 20 STO 03  

 21 CLX  

 22▪LBL 02  

 23 XEQ 06 

 24 RCL 01 
 25 +  

 26 RCL 02 

 27 *  

 28 DSE 03 

 29 GTO 02 

 30 "SUM=" 
 31 GTO 07 

 32▪LBL 05  

 33 PROMPT 

 34 STO 01 

 35 "I%=?" 

 36 PROMPT 
 37 1200  

 38 /  

 39 1  

 40 +  

 41 STO 02 

 42 FIX 0 

 43 CF 29 

 44 RTN  

 45▪LBL 06  
 46 RCL 03 

 47 STO 04 

 48 RDN  

 49 12  

 50 ST/ 04  

 51 X<> 04 
 52 FRC  

 53 X=0?  

 54 VIEW L 

 55 RDN  

 56 RTN  

 57▪LBL "LOAN" 
 58 "AMT=?"  

 59 XEQ 05  

 60 "NR.M=?" 

 61 PROMPT  

 62 STO 03  

 63 Y^X  

 64 ,  

 65▪LBL 03  

 66 XEQ 06 
 67 RCL 02  

 68 *  

 69 1  

 70 +  

 71 DSE 03  

 72 GTO 03  
 73 /  

 74 *  

 75 "M.P="  

 76▪LBL 07  

 77 FIX 2  

 78 ARCL X  
 79 FIX 5  

 80 SF 29  

 81 AVIEW  

 82 END 

 

(177 bytes)

Registers, Labels and Flags 

REGISTERS COMMENTS  LABELS COMMENTS 

R01 Loan amount or monthly payment  LBL00 Main entry 

R02 Annual interest  LBL02 Loop around savings 

R03 Duration in number of months  LBL03 Loop around loan 

R04 Work register for count down  LBL05 Common stop for interest 

   LBL06 Calculate year count down 

   LBL07 Reset default, show result 

  

FLAGS COMMENTS 

29 Type of thousands separator 

 

 

Downloads 

The RAW/TXT format of the program is available via the website: LOAN+SAVE (in zip file). 

 

https://hooxies.nl/?page_id=29696

